

Power Series and Taylor/Maclaurin Series

Douglas Meade, Ronda Sanders, and Xian Wu

Department of Mathematics

Overview

There are two fundamental questions to ask about a power series $\sum_{k=0}^{\infty} c_k x^k$ (or $\sum_{k=0}^{\infty} c_k (x - x_0)^k$):

1. For what values of x does the infinite sum converge?
2. When the series converges, to what function does it converge? Notice that, in this case, the series becomes the Taylor series of the function.

The first question is answered by finding the radius of convergence using the ratio test for absolute convergence and then check at endpoints if needed. This can be easily carried out by hand, but maple can be used to help with computations.

The second question is, in general, much more difficult to answer. One useful technique is to apply common operations (substitution, multiplication, division, differentiation, integration, ...) to a known power series, such as the Taylor series of some basic functions listed below. Maple is much better at this than most of us, but a little practice can quickly improve this skill.

One important application of power series is to approximate a function using partial sums of its Taylor series. The accuracy of the approximation depends on the size of the interval called for and the number of terms used in the partial sum. Maple can help us to visualize those with simple animations.

Related Course Material/Preparation

Calculus Text: §11.8 to §11.10. Maple Text: §9.5. Be sure to review steps to find radius and interval of convergence for a power series. Also, be able to write down – from memory – the following basic Taylor series (and their intervals of convergence):

$$1) \quad \frac{1}{1-x} = \sum_{k=0}^{\infty} x^k, \quad -1 < x < 1$$

$$2) \quad e^x = \sum_{k=0}^{\infty} \frac{x^k}{k!}, \quad -\infty < x < \infty$$

$$3) \quad \sin(x) = \sum_{k=0}^{\infty} \frac{(-1)^k x^{2k+1}}{(2k+1)!}, \quad -\infty < x < \infty$$

$$4) \quad \cos(x) = \sum_{k=0}^{\infty} \frac{(-1)^k x^{2k}}{(2k)!}, \quad -\infty < x < \infty$$

$$5) \quad \ln(1+x) = \sum_{k=0}^{\infty} \frac{(-1)^k x^{k+1}}{k+1}, \quad -1 < x \leq 1$$

Maple Essentials

- New Maple command introduced in this lab include:

Command/Example	Description
<code>display(L,insequence=true,options);</code> Example: <code>display(seq(plot([x^k,f],x=-1..1),</code> <code>k=0..8),insequence=true);</code> <code>display(seq(plot(k*x,x=0..1,y=0..8),</code> <code>k=1..8),insequence=true);</code>	Displays a collection of plots L as the frames of an animation, where L can be a set, list, or array of PLOT structures. Standard plot options can be used. – without <code>insequence=true</code> all frames are displayed in one plot – the <code>display</code> command is in the <code>plots</code> package

Activities

1. Consider the power series $\sum_{k=0}^{\infty} \frac{x^{k+1}}{(k+1)5^{k+1}}$.
 - (i) Find the interval of convergence (and radius of convergence) of this series.
 - (ii) To what function does this series converge?
 - (iii) Animate approximations of the function with the first 8 partial sums of its Taylor series over two intervals of different size. What have you learned from the animations?
2. Repeat Activity 1 for each of the following power series:

(a) $\sum_{k=0}^{\infty} x^{2k}$	(b) $\sum_{k=0}^{\infty} \frac{x^{k+1}}{k!}$	(c) $\sum_{k=0}^{\infty} \frac{(-1)^k x^{2k}}{(2k+1)!}$
(d) $\sum_{k=0}^{\infty} kx^{k-1}$	(e) $\sum_{k=0}^{\infty} k(k-1)x^{k-2}$	(f) $\sum_{k=0}^{\infty} (x-1)^k$

Example: Activity 1.

> with(plots):

Note: First, we define the k -th term a_k , the series, and the partial sum. Notice that the command `Sum` is used instead of `sum`, as we don't want to evaluate them at this time.

> a:=k->x^(k+1)/((k+1)*5^(k+1));

> S:=Sum(a(k),k=0..infinity);

> PS:=Sum(a(k),k=0..n);

Note: Next, we apply the ratio test for absolute convergence and then solve for the range of convergent x .

> r:=abs(a(k+1)/a(k));

> r:=simplify(r);

> rho:=limit(r,k=infinity);

> R:=solve(rho<1,x);

Note: So the radius of convergence is 5. The following commands check at the endpoints of the interval.

> AtFive:=simplify(eval(S,x=5));

Note: It is the harmonic series so it should diverge.

> value(AtFive); # Or Right-click and Evaluate.

> AtNegFive:=simplify(eval(S,x=-5));

Note: Easy to see it should converge by the alternating series test.

> value(AtNegFive); # Or Right-click and Evaluate.

Note: So the power series diverges at endpoint $x = 5$ but converges at endpoint $x = -5$. The interval of convergence is hence $[-5, 5)$. Our next step is to see to what function it converges.

> f:=value(S);

Note: Does the result remind you something? In fact, we can see that S is just the Taylor series of the above function (use formula 5 for $-\ln(1+x)$ and replace x by $-x/5$). Finally, the following commands animate approximations of the function with the first 8 partial sums over given intervals.

> q1:=seq(PS, n=0..7);

> value(q1);

> q2:=seq(plot([p,f],x=-2..2),p=q1):

> display(q2,insequence=true);

> q3:=seq(plot([p,f],x=-4..4),p=q1):

> display(q3,insequence=true);

Note: We can see that the approximations worked very well, but more terms are needed as the interval is getting bigger. (You may also want to try maple's **Taylor Approximation** tutor in **Calculus-Single Variable** from **Tools** menu.)

Assignment

There is no assignment this week so you will have more time to review Lab 7 to Lab 11 for Quiz 2 next week